

NATIONAL RADIO ASTRONOMY OBSERVATORY
SOCORRO, NEW MEXICO*

ELECTRONICS DIVISION INTERNAL REPORT No. 172

WAVEGUIDE MOUNTING POST PROGRAM

STEVE MAAS

NOVEMBER 1976

NUMBER OF COPIES: 150

*Request for additional copies should be sent to:
NRAO Electronics Division
P. O. Box 2
Green Bank, West Virginia 24944

WAVEGUIDE MOUNTING POST PROGRAM

Steve Maas

TABLE OF CONTENTS

	Page
1. Introduction	1
2. Limitations.	1
3. Program Description and Operation.	3
4. Verification	3
5. References	4

ILLUSTRATIONS

Figure 1. Mounting Post Geometry.	5
Figure 2. Waveguide Mounting Post Program	6

WAVEGUIDE MOUNTING POST PROGRAM

1. INTRODUCTION

The purpose of this program is to calculate the input impedance to a gap in a cylindrical transversally mounted waveguide post. It is expected that such a structure, shown in Figure 1, would be used for mounting a two-terminal electronic device in a rectangular waveguide, and its accurate analysis would be helpful in designing a wide variety of microwave components.

The analysis of this structure is presented in a paper by Eisenhart and Khan. This paper gives the first rigorous analysis of this structure, and excellent agreement between theory and measured impedances is indicated. Moreover, the method is useful at all frequencies, from below cutoff to frequencies at which many modes can propagate. Consequently it is particularly useful in mixer and parametric amplifier analysis, where the termination presented to the diode at harmonics of the signal and pump frequencies, and their mixing products, is critical to the performance of the component.

2. LIMITATIONS

The program is written in the BASIC language for use on the Hewlett-Packard 9830 programmable calculator, but can be adapted to other computers with little change. In order to take advantage of the simplicity and convenience of the programmable calculator, several limitations to the analysis are necessary:

- (a) The frequency and waveguide dimensions must be chosen so that only one mode propagates in the waveguide: the dominant TE_{10} mode. This limitation eliminates the advantage of analyzing the gap impedance over a wide bandwidth, as mentioned above, but is necessary when calculations are made with a language which cannot

easily handle complex numbers. Programming the equations in Fortran would solve the problem, at the cost of the convenience of using the programmable calculator.

- (b) It is assumed that one end of the waveguide is connected to a matched termination and the other end is terminated in a short circuit an odd number of quarter wavelengths from the post. It is also assumed that the short is far enough from the post that non-propagating modes are not affected by it, and close enough so that the TE_{10} mode is not significantly attenuated.
- (c) Only a limited number of modes can be considered. The effects of $TE_{m,n}$ and $TM_{m,n}$ modes up to $m=20$ and $n=12$ are calculated. The practical result of this limitation is that only certain values of post diameter and gap width can be used, within a given accuracy. For 1% maximum error,

$$.25a > d > \frac{2a}{m}$$

and
$$.25b > g > \frac{2b}{n}$$

where a and b are the waveguide height and width, and d and g are the post diameter and gap width, respectively.

These relations specify a fairly narrow range of post and gap dimensions. However, a gap width or post diameter one-third of the minimum shown above will result in an error of less than 12%. This error is still well below the error inherent in the design of many kinds of microwave components, and piece to piece variations in solid state device parameters.

3. PROGRAM DESCRIPTION AND OPERATION

The program is divided into three parts. The first collects the data (waveguide, post, and gap dimensions, and frequency) and prints a warning if the gap and post dimensions exceed the limitations for a maximum 12% error. The second part calculates the "mode pair impedances", the combined contribution to the input impedance by TE and TM modes of the same mode indices (m,n). These are calculated sequentially and stored in a two dimensional array, and in the third part, they are combined in series and parallel as necessary to determine the gap input admittance.

During the second part, the message "I'm working on it!", followed by the present value of n, is displayed after each iteration through the mode pair impedance "for-next" loop. In the third part, the message "Hold your breath" is displayed, followed in the same way by n. The purpose of these displays is not so much whimsy as to let the user know that the program is progressing successfully, and to dispel fears arising during the long (~5 minutes) execution that the program is trapped in an unending loop.

Operation of the program is straightforward. The operator must enter the waveguide and post dimensions in inches, and the frequency in gigahertz. The dimensions are defined in Figure 1.

The gap input admittance is printed in the output. Conversion to impedance is trivial. The waveguide and post dimensions are reprinted as a permanent record.

4. VERIFICATION

In order to verify that the program is correct, several of the results published by Eisenhart and Khan were recalculated with this program. All the results agreed within the errors predicted above (some corrections were necessitated by the fact that doubly terminated waveguides were used in the original paper).

5. REFERENCES

Eisenhart, R.L., and Khan, P.J., "Theoretical and Experimental Analysis of a Waveguide Mounting Structure", IEEE Trans. on Microwave Theory and Techniques, MTT-19, No. 8, pp 706-719, August 1971.




Figure 1: MOUNTING POST GEOMETRY.

- a = waveguide width
- b = waveguide height
- d = post diameter
- s = distance from wall
- g = gap width
- h = height to center of gap

```

4 DISP "ENTER ALL DIMENSIONS IN INCHES"
8 WAIT 5000
12 PRINT "DIMENSIONS IN INCHES"
16 DISP "WAVEGUIDE WIDTH, HEIGHT":
20 INPUT A[1],B[1]
24 LET A=0.0254*A[1]
28 LET B=0.0254*B[1]
32 DISP "POST DIA,DIST FROM WALL":
36 INPUT W[1],S[1]
40 LET W=0.0254*W[1]
44 LET S=0.0254*S[1]
48 LET W[2]=W/A
52 DISP "GAP WIDTH, HEIGHT TO CENTER OF GAP":
56 INPUT G[1],H[1]
60 LET G=0.0254*G[1]
64 LET H=0.0254*H[1]
68 LET G[2]=G/B
72 IF W[2]>0.25 THEN 84
76 IF G[2]>0.25 THEN 84
77 IF W[2]<0.0333 THEN 84
78 IF G[2]<0.0556 THEN 84
80 GOTO 96
84 PRINT "DIMENSIONS OF THE GAP EXCEED THE LIMITS OF THE MODEL"
88 PRINT
92 PRINT
96 DISP "FREQ, GHZ":
100 INPUT F[1]
104 REM CALCULATE INITIAL PARAMETERS
108 LET F=F[1]*10^9
112 PRINT "WAVEGUIDE HEIGHT",B[1]"WIDTH",A[1]
116 PRINT "POST DIA",W[1],"DIST FROM WALL",S[1]
120 PRINT "GAP WIDTH",G[1],"HEIGHT TO CTR OF GAP",H[1]
124 PRINT "FREQUENCY, GHZ",F[1]
128 PRINT
132 LET L=0.3/F[1]
136 IF A>L THEN 350
138 IF A<L/2 THEN 350
140 LET P=3.14159
144 LET K=2*P/L
148 REM CALCULATE MODE PAIR IMPEDANCES AND POST COUPLING FACTORS
152 DIM J[22],K[15],C[22],E[15],Z[22,15]
156 LET E[1]=1
160 FOR N=0 TO 12
162 DISP "I'M WORKING ON IT!"*N*":....."
163 WAIT 500
164 LET K[N+1]=N*P/B
168 IF N=0 THEN 176
172 LET E[N+1]=COS(K[N+1]*H)*(SIN(N*P*G[2]/2)/(N*P*G[2]/2))
176 FOR M=1 TO 20
178 LET J[M]=M*P/A
179 LET C[M]=SIN(J[M]*S)*(SIN(M*P*W[2]/2)/(M*P*W[2]/2))
180 IF M=1 AND N=0 THEN 188
182 LET Z[M,N+1]=377*B*(K^2-K[N+1]^2)/(2*A*K*(J[M]^2+K[N+1]^2-K^2)+0.5)
184 GOTO 192
188 LET Z[M,N+1]=754*B*K/(A*(K^2-J[M]^2)+0.5)
192 NEXT M
196 NEXT N

```

Figure 2: WAVEGUIDE MOUNTING POST PROGRAM


```

200 LET N=0
204 LET M=1
208 LET Z=0
212 LET X[1]=0
216 LET Y=0
220 FOR N=0 TO 12
222 DISP "HOLD YOUR BREATH!" ; N ; "....."
223 WAIT 300
224 LET Y[1]=0
226 LET T=2
228 FOR M=T TO 20
240 IF N=0 THEN 252
244 LET X[1]=X[1]+(Z[M,N+1]*(C[M]2/E[N+1]2))
248 GOTO 256
252 LET Z=Z+Z[M,1]*C[M]2
254 LET T=1
256 NEXT M
260 IF N#0 THEN 280
264 LET Z[1,1]=Z[1,1]*C[1]2
268 LET R=Z[1,1]/(Z[1,1]2+Z2)
272 LET Y[1]=Z/(Z[1,1]2+Z2)
276 GOTO 284
280 LET Y[1]=1/X[1]
284 LET Y=Y+Y[1]
288 NEXT N
292 PRINT "POST DRIVING POINT ADMITTANCE, REAL PART:",R
296 PRINT "IMAGINARY PART: J*";(-1)*Y
300 FOR I=1 TO 10
304 BEEP
306 WAIT 500
308 NEXT I
312 STOP
350 PRINT "WAVEGUIDE CUT OFF OR MULTI-MODED"
354 STOP
358 END

```